﻿Public Class Form1
  Private Sub Button1_Click(
      ByVal sender As System.Object,
        ByVal e As System.EventArgs) _
        Handles Button1.Click
    Dim indirizzi As XDocument =
       <?xml version="1.0"
       encoding="utf-16" standalone="yes"?>
       <Contatti>
         <Contatto id="C01">
           <nome>Mario</nome>
           <cognome>De Ghetto</cognome>
           <riferimenti>
             <indirizzo tipo="email">
               mario@deghetto.it
             </indirizzo>
             <indirizzo tipo="url">
               http://deghetto.wordpress.com
             </indirizzo>
           </riferimenti>
         </Contatto>
         <Contatto id="C02">
           <nome>Giuseppe</nome>
           <cognome>Verdi</cognome>
           <riferimenti>
             <indirizzo tipo="email">
               giuseppe.verdi@teatro.it
             </indirizzo>
             <indirizzo tipo="url">
               http://www.giuseppeverdi.it
             </indirizzo>
           </riferimenti>
         </Contatto>
       </Contatti>
    indirizzi.Save("indirizzi.xml")

    ' variabili locali per messaggio
    Dim messaggio As String = ""
    Dim NL As String = Environment.NewLine
    ' interrogazione
    Dim doc = XDocument.Load("indirizzi.xml")
    Dim Contatti =
      From c In doc...<Contatto>
      Where c.@id = "C02"
      Select New With {
        .Nome = c.Element("nome").Value,
        .Cognome = c.Element("cognome").Value}
    For Each Contatto In Contatti
      messaggio &= "Il contatto si chiama " &
      Contatto.Nome & " " & Contatto.Cognome & NL
    Next
    MessageBox.Show(messaggio)

  End Sub
End Class
